Streuung am Potentialtopf

Notation im Vergleich zu Abschnitt 2.3 der Vorlesung:
y=x/L v=(2mL^2/hbar^2)V_0 e=(2mL^2/hbar^2)E
in y reicht das Potential also von −1 bis 1

In[1]:=
```
(* define WFs in part 1,2,3 *)
A[a_, b_] := Cos[2 b] - I / 2 (b / a + a / b) Sin[2 b];
B[a_, b_] := I / 2 (b / a - a / b) Sin[2 b];
psi1[y_, e_, v_] :=
  (A[a, b] Exp[I a (y + 1)] + B[a, b] Exp[-I a (y + 1)]) /. {a → Sqrt[e], b → Sqrt[e + v]};
psi2[y_, e_, v_] := (1 / 2 (1 + a / b) Exp[I b (y - 1)] + 1 / 2 (1 - a / b) Exp[-I b (y - 1)]) /.
   {a → Sqrt[e], b → Sqrt[e + v]};
psi3[y_, e_, v_] := Exp[I Sqrt[e] (y - 1)];
(* define total wave fct normalized to 1 for incoming coeff *)
psi[y_, e_, v_] := (UnitStep[-1 - y] psi1[y, e, v] +
     UnitStep[1 - y^2] psi2[y, e, v] + UnitStep[y - 1] psi3[y, e, v]) / A[Sqrt[e], Sqrt[e + v]];
(* plot transmission probability T(E) *)
emax[v_] := (Floor[2 / Pi Sqrt[v]] + 3.2)^2 Pi^2 / 4 - v; (* fit about 3 resonances in plot *)
Manipulate[Plot[1 / (1 + Abs[B[Sqrt[e], Sqrt[e + v]]]^2),
  {e, 0, emax[v]}, PlotRange → {0, 1}], {v, 0.1, 30}]
(* plot absolute square of WF *)
Manipulate[Plot[Abs[psi[y, e, v]]^2, {y, -4, 2}, PlotRange → {0, 2}],
 {v, 0.1, 30}, {e, 1, emax[v]}]
(* plot WF *)
Manipulate[ParametricPlot3D[{y, Im[psi[y, e, v]], Re[psi[y, e, v]]}, {y, -4, 4}],
 {v, 0.1, 30}, {e, 1, emax[v]}]
```

Out[8]=



Out[9]=



Out[10]=